MPR

MPR Associates, Inc.
320 King Street
Alexandria, VA 22314

CALCULATION TITLE PAGE

Client GPU Nuclear	Page 1 of ~~19~~ 17 + Appendices (BRL, CBS, WMcC Rev. 1)
Project Shroud Vertical Weld Evaluation	Task No. 083-9601-248-0
Title Shroud Finite Element Evaluation	Calculation No. 083-248-CBS-01

Preparer/Date	Checker/Date	Reviewer/Approver Date	Rev. No.
C. Swanner Craig B Swanner 5/26/98	B. Lane Ben Lane 5/26/98	W. McCurdy William McCurdy 5/27/98	0
Ben Lane 7/16/98	Craig B Swanner 7/16/98	William McCurdy 7/16/98	1

QUALITY ASSURANCE DOCUMENT

This document has been prepared, checked, and reviewed in accordance with the Quality Assurance requirements of 10CFR50 Appendix B, as specified in the MPR Quality Assurance Manual.

MPR

MPR Associates, Inc.
320 King Street
Alexandria, VA 22314

RECORD OF REVISIONS

Calculation No.	Prepared By	Checked By	Page 2
083-248-CBS-01	Ben Love	C Swanson	

Revision	Description
0	Initial Issue
1	Revised seismic loads based on updated transient dynamic analyses. Removed results for 10 wedges because only 8 wedges will be installed.

1.0 PURPOSE

The purpose of this calculation is to determine the flaw tolerance of the vertical welds in the section between circumferential welds H5 and H6A in the Oyster Creek core shroud. The vertical weld evaluation was originally performed in Reference 1. This calculation considers the effect of installing wedges between the core plate and the shroud wall. A finite element model of the shroud section is developed to evaluate the effects.

The finite element model is also used to determine the leakage path flow area through the cracked vertical weld during normal operating conditions.

2.0 SUMMARY OF RESULTS

The maximum stresses in the shroud section between circumferential welds H5 and H6A are summarized for the limiting load cases in Table 2-1. Stress contours for each load case are presented later in this calculation. As shown, these stresses meet the requirements of Subsection NB of the ASME Boiler and Pressure Vessel Code, 1989 Edition. The evaluations are performed with eight core plate wedges installed. All circumferential welds and the vertical welds in the H5/H6A shroud section are assumed to be completely failed. The evaluations show that the load through the vertical welds can be reacted by taking credit for compression across the failed circumferential welds due to tie rod preload.

For the MSLB case, if only welds H5 and H6A are failed with all other circumferential welds intact, compression could no longer be maintained across both welds H5 and H6A. Consequently, some amount of the vertical weld is required to react the hoop load from the differential pressure. Results of the evaluation performed in Appendix A show that if there is ten inches of intact vertical weld, the stresses in the H5 and H6A meet the requirements of the ASME Code.

The maximum leakage path flow area through a fully-cracked vertical weld in the H5/H6A shroud segment during normal operating conditions is 4.67 in^2. This flow area will be used elsewhere to evaluate the effect of reactor coolant flow that bypasses the core through the cracked vertical weld.

Table 2-1

Summary of Maximum Stresses in the Shroud Shell Between Circumferential Welds H5 and H6A

Stress Type	Stress Limit	Calculated Stress (ksi)	Allowable Stress (ksi)	Stress Ratio
Service Level D - Main Steam Line Break plus Safe Shutdown Earthquake (Note 1)				
Primary Membrane plus Bending (Pm+Pb)	3.6 Sm	31.7	60.0	0.53
Service Level D - Recirculation Line Break plus Safe Shutdown Earthquake				
Primary Membrane plus Bending (Pm+Pb)	3.6 Sm	44.5	60.0	0.74

Note:

1. The calculated stresses tabulated are for the locations at the wedges. The maximum stress shown in the stress contours presented in Section 4 are at the intersection of horizontal weld H5 and the vertical weld. These stresses are considered secondary since they are a result of the structural discontinuity at this location. Consideration of secondary stresses is not required for Service Level D loading.

3.0 DISCUSSION

3.1 Shroud Configuration

The design function of the core shroud is to provide lateral support for the nuclear fuel and to provide a flow partition within the reactor vessel. A developed view of the Oyster Creek reactor is shown in Figure 3-1. The shroud is divided into a number of cylindrical section by its circumferential welds designated H1 through H7. Each of these shroud sections has two vertical welds which join individual rolled plates to form the cylinders.

The shroud configuration is further modified by installation of the core shroud repair. The core shroud repair consists of ten tie rod restraint assemblies that structurally replace all the circumferential welds (H1 through H7). Figure 3-1 shows the location and positioning of each tie rod assembly. Note that each shroud segment between two adjacent circumferential welds has radial restraints to react lateral loads.

3.2 Applied Loads

The following applied loads are reacted by the shroud cylinders.

- *Differential Pressure*: The pressure difference across the shroud creates a hoop load (stress) in the shroud which must be reacted through vertical welds in each shroud cylinder. The differential pressures from Appendix A of Reference 5 are used in this calculation.

- *Seismic Bumper Loads*: The reactor vessel, shroud, and fuel are excited in the horizontal direction by an earthquake. Relative motion between the shroud and reactor vessel causes the radial restraints to contact the shroud resulting in lateral loads in the shroud shell.

- *Seismic Fuel Loads*: The reactor vessel, shroud, and fuel are excited in the horizontal direction by an earthquake. Motion of the fuel causes lateral loads normally reacted through the core support ring and the top guide.

- *Lateral RLB Loads*: A recirculation line break (RLB) causes lateral loads on the shroud shell. Lateral loads are transmitted into the shroud barrel when the shroud contacts the radial restraints.

During accident events, lateral loads are transmitted from the fuel to the core plate. The original plant load path is through the core plate holddown bolts into the core support ring (between H6A and H6B) and out through the shroud. The core shroud repair structurally replaces all shroud horizontal welds with an alternative load path. For lateral loads, the alternative load path used in the analysis (Reference 2) was through the core support ring and directly into seismic restraints into the reactor vessel.

Oyster Creek plans to install eight wedges between the core support plate and the shroud cylinder between horizontal welds H5 and H6A. The wedges are designed to bypass the core plate holddown bolts thereby precluding the need to inspect and maintain the bolt preload. The wedges react the lateral fuel load from the core plate directly into the shroud. This alternative load path was not analyzed as part of the original flaw tolerance evaluation or the core shroud repair design.

3.3 Calculation Method

The acceptable flaw lengths in the vertical welds of the Oyster Creek core shroud were determined in Reference 1. The evaluation was based on the shroud loadings and load paths used in the analysis of the core shroud repair (Reference 2). The evaluation used a limit load approach by determining the amount of weld length required to maintain the limiting stresses within the ASME Code limits.

This calculation develops a finite element model to analyze the shroud cylinder between welds H5 and H6A with the core plate wedges installed. Specifically, this calculation will address the effect of potential cracking in the vertical welds on the structural integrity of the shroud cylinder between H5 and H6A. In all evaluations, both H5 and H6A are assumed to be cracked through wall, all the way around the shroud. This is the most limiting condition for the vertical welds because intact ligament in the horizontal weld provides an alternate load path around the vertical welds for hoop loads in the shroud cylinder.

3.4 Service Loadings

The core shroud repair is designed to react loads arising from the following shroud service conditions (see Reference 5).

Table 3-1. **ASME Code Service Limits**

Load Case	ASME Service Limit	Load Combination	ASME Allowable Primary Membrane Stress Intensity (Note 1)
Upset	Level B	Upset differential pressure (ΔP)	Sm
OBE	Level B	Operating basis earthquake (OBE) loads plus normal ΔP	Sm
SSE + MSLB	Level D	Main steam line break (MSLB) ΔP plus safe shutdown earthquake (SSE) loads	Lesser of 2.4 Sm or 0.7 Su
SSE + RLB	Level D	Recirculation line break (RLB) loads plus normal ΔP plus SSE loads	Lesser of 2.4 Sm or 0.7 Su

Notes:

1. The allowable stresses are from Section NB-3220 of Reference 3. Specifically, the limits for design loads from NB-3221 are applied to Level B loads, and the limits from NB-3225 and F-1331 are applied to Level D loads. Primary membrane plus bending stress limits are 1.5 times the primary stress limit. Also, note that Sm is the allowable stress of the material at design temperature and Su is the ultimate tensile strength of the material at design temperature.

The controlling service loadings for comparison with the stress limits can be determined by examining the load components for each service condition. From Reference 4, the RLB load case is most limiting for the H5/H6A shroud segment in terms of required vertical weld. The differential pressure associated with the MSLB case also causes vertical loads in the shroud large enough to separate potentially cracked horizontal welds. This results in different boundary conditions for the H5/H6A shroud section. As a result, this calculation investigates only the MSLB + SSE and RLB + SSE load cases.

MPR		MPR Associates, Inc. 320 King Street Alexandria, VA 22314	
Calculation No. 083-248-CBS-01	Prepared By Ben Lane	Checked By C Arvan	Page 8



Figure 3-1. Reactor Developed View

4.0 FINITE ELEMENT MODEL

The input for each model used in this analysis are contained in the Appendix C.

4.1 Loading Conditions

Loading conditions for the evaluation of the H5/H6A shroud cylinder are summarized below in Table 4-1.

Table 4-1. **Loading Conditions**

Loading	Value	Reference
MSLB + SSE:		
- Steam line break differential pressure across shroud wall	19.0 psi	5
- SSE seismic bumper load	74 kips	(Note 1)
- SSE seismic fuel load	40 kips	(Note 1)
RLB + SSE:		
- Normal operating differential pressure across shroud wall	4.34 psi	5
- SSE seismic bumper load	74 kips	(Note 1)
- SSE seismic fuel load	40 kips	(Note 1)
- Lateral recirculation line break load	41.3 kips	(Note 2)

Notes:

1. The most limiting case for the vertical welds in the H5/H6A shroud section is when both horizontal welds are broken. The seismic load case corresponding to this condition is the multiple weld break case. The maximum bottom bumper load and maximum bottom fuel load for the multiple break case are taken from Reference 6.

2. The lateral load due to an RLB is determined from the shroud shear distribution given in Reference 7. The load applied to the H5/H6A shroud section is calculated as the difference in shear loads at horizontal welds H5 and H6A.

4.2 Material Properties

The material properties of the shroud are summarized in Table 4-2 below. These properties are obtained from Appendix I of Reference 3 at the design temperature of 575°F (Reference 5).

Table 4-2. **Material Strength Properties**

Component	Material	Material Properties	
		Property	Value
Shroud	SA 240, Type 304 (Reference 8)	Allowable Stress (Sm) Yield Stress (Sy) Ultimate Strength (Su) Modulus of Elasticity (E)	16.675 ksi 18.5 ksi 63.5 ksi 25,400 ksi

Note that a Poisson's ratio (ν) of 0.3 and a density of 0.29 lb_m/in^3 is assumed for all materials in this calculation.

4.3 Model Geometry

To evaluate the vertical welds in the H5/H6A shroud cylinder, the shroud sections between horizontal welds H4 and H6B are modeled. The additional sections are modeled to include the effect of compression across horizontal welds from tie rod preload in the evaluation of the vertical welds. The core plate wedges and the radial restraints are also included in the finite element model. The geometric data used to construct the finite element model of the shroud is shown in Figure 4-1. The three-dimensional, finite element model of the shroud is shown in Figure 4-2.

The shroud is modeled with ANSYS SOLID45 elements. These are 8-node, brick elements with three displacement degrees of freedom at each node. Bearing between the shroud and the core plate wedges and between the shroud and the radial restraints is modeled by coupling the bearing surface to the shroud. Both vertical welds in the shroud section are assumed to be completely cracked.

4.4 Loads and Boundary Conditions

Three individual ANSYS runs for the two load cases are performed for this evaluation. For each load case (MSLB + SSE and RLB + SSE), the differential pressure is run separately from the lateral loads. The results of the two runs are combined by superposition to determine the net state of stress in the shroud.

For the differential pressure cases, the differential pressure is applied to all nodes on the inside surface of the shroud. For the lateral load case, the seismic fuel load is assumed to be directed at the vertical weld located at 165°. This results in the highest shroud shell bending stresses locally near the vertical weld.

The seismic fuel load is reacted to the shroud through the wedges. The core plate is assumed to be infinitely rigid. This is modeled by coupling all the wedges to a node at the center of the shroud. The seismic fuel load is then applied to the center node and directed toward the vertical weld at 165°. It is assumed that there is no gap between the shroud and the wedges. Accordingly, the wedges are directly coupled in all degrees of freedom to the shroud.

The seismic bumper load is conservatively assumed to be directed at the vertical weld at 165°. Nominally, there is a 0.375" total gap between the radial restraints, the shroud outside diameter, and the reactor inside diameter. Because the restraints are located at various positions around the shroud, only several restraints will carry load when the shroud is loaded in a given direction. Initial scoping analyses have been performed to show that only the three restraints closest to the vertical weld are loaded when the lateral load is applied in the 165° direction. In this analysis, it is assumed that there is no gap between the restraints and shroud. This is modeled by directly coupling the restraints to the shroud in all degrees of freedom.

The RLB lateral load is directed toward the recirculation nozzle with the faulted pipe. This analysis conservatively assumes that the RLB lateral load is directed at the vertical weld at 165°. Since the seismic bumper load and RLB lateral load are assumed to act in the same direction, they are applied to the model as a combined uniform acceleration. The acceleration is determined by subtracting the seismic fuel load from the total lateral load (seismic bumper plus RLB lateral load) and then dividing by the total model mass.

The following model boundary conditions are applied to all the models:

- The vertical welds in the H5 and H6A shroud section are assumed to be completely failed. Since the welds tend to open up due to pressure, this is modeled by uncoupling the nodes on either side of the weld.

- All circumferential welds are considered completely failed. The tie rod preload keeps the entire shroud in compression during normal operation. The differential pressure during an RLB is bounded by normal operating conditions. Consequently, there is compression across all the failed circumferential welds during the RLB + SSE load case. During a MSLB accident the differential pressure across the shroud is significantly increased. This results in a large upload applied to the core plate due to differential pressure. Since it is assumed a weld below the core support ring is failed, the large core plate upload and the restraining loads from the tie rods keep all the circumferential welds above the core support ring in compression. Since there is compression across each of the failed circumferential welds, H5 and H6A are modeled as pinned joints, i.e. they can carry shear but not moment. (Note that if only H5 and H6A are failed during an MSLB event, the H5/H6A shroud segment becomes free.

This case is examined more closely in Appendix A.)

The pinned joints are modeled by coupling the nodes on the inside diameter of the weld of the two adjacent cylinders in all three translational directions. The circumferential welds at H4 and H6B are away from the analysis area of interest and are considered free (i.e., no applied boundary conditions).

- The model is restrained vertically at the node in the center of the core plate.
- The outside surface of the radial restraints are restrained from motion in the radial direction.

4.5 Finite Element Analyses Results

For each loading condition, the stresses are evaluated using a finite element model of the shroud (see Section 3.5). Stress analysis results are obtained using finite element methods and the ANSYS finite element program. The ANSYS evaluations are performed on a Sun UltraSPARC Workstation with the Solaris 2.5 operating system. The ANSYS installation verification was performed in QA-53-3.

The finite element model of the H5/H6A shroud section is evaluated for loading conditions described in the previous sections. Analysis results for the SSE + MSLB and the SSE + RLB load cases are shown in Figures 4-3 and 4-4. Stress results are summarized and compared to the allowable stresses in Table 2-1.

Figure 4-1. Finite Element Model Geometry
(References 1, 2, 8, 9, and 10)



ANSYS 5.3
APR 24 1998
14:46:18
PLOT NO. 1
ELEMENTS
TYPE NUM

XV =1
YV =1.5
ZV =1
DIST=157.557
XF =1.378
ZF =71.448
A-XM=-90
PRECISE HIDDEN

- Notes:

1) Missing connection lines between nodes are a result of difficulty by ANSYS in identifying hidden surfaces. Nodes not shown are present in the model.
2) Node spacing is reduced at wedge azimuths to provide greater accuracy in determing stress distributions.



Figure 4-2. Finite Element Model of the Shroud



Figure 4-3. Stress Contour for MSLB + SSE

MPR

MPR Associates, Inc.
320 King Street
Alexandria, VA 22314

Calculation No.	Prepared By	Checked By	
083-248-CBS-01	Ben Lane	C Swan	

Figure 4-4. Stress Contour for RLB + SSE



5.0 REFERENCES

1. MPR Report 1762, "Oyster Creek Nuclear Generating Station, Evaluations of the Upper Shroud Ledge and Shroud Vertical Welds," October 1996, Revision 0.

2. MPR Report, "Oyster Creek Nuclear Generating Station, Core Shroud Repair, Design Report," October 1994, Revision 1 (Two Volumes).

3. ASME Boiler and Pressure Vessel Code, Section III, Subsection NB, 1989 Edition.

4. MPR Calculation 083-224-03, "Required Intact Weld Length for Vertical Welds Based Upon a Limit Load Analysis," Revision 0.

5. MPR Specification 083-9403-001, "Design Specification for Oyster Creek Nuclear Generating Station (OC) Core Shroud Repair," Revision 0.

6. MPR Calculation 083-261-BRL-2, "Transient Dynamic Evaluation of OC Shroud with Vertical Welds Failed," Revision 0.

7. MPR Calculation 083-205-13, "Tie Rod Assembly Loads for Recirculation Line Break," Revision 0.

8. GE Drawing 105E1413B, "Oyster Creek, Shroud Data," Sheet 1, Revision 1.

9. GPUN Letter E520-98-008 from A. Collado to P. Kasik (MPR), "Oyster Creek Core Shroud Vertical Welds Evaluation with Core Plate Wedges," Dated February 18, 1998.

10. GPUN Drawing 3E-222-29-1002, "Reactor Vessel Shroud, 16R Inspection Report," Revision 1.

APPENDIX A

MSLB + SSE Load Case
With Only H5 and H6A Failed

A.1 PURPOSE

The purpose of this appendix is to evaluate the effect of failed vertical welds in the H5/H6A shroud segment for the MSLB + SSE load case if only circumferential welds H5 and H6A are completely failed. This appendix accounts for the effect of the core plate wedges installed between the core plate and the shroud shell between H5 and H6A.

A.2 RESULTS

If any ten inches of both vertical welds are intact, then the stresses in the H5/H6A shroud segment meet the requirements of the Subsection NB of the ASME Code during the MSLB + SSE load case.

A.3 FINITE ELEMENT MODEL

The H5/H6A section of the finite element model described in the main body of this calculation is used in this evaluation. The differences in loading and boundary conditions are described below:

- During a MSLB, the differential pressure loads are sufficient to overcome the tie rod preload. If only welds H5 and H6A are failed, the shroud could potentially separate at both welds H5 and H6A. As a result, the assumption, used in the main body of the calculation, that there is compression across failed circumferential welds is no longer valid. This condition is considered in this appendix by only modeling the shroud section between H5 and H6A. The shroud is given no constraint at the failed welds.

- Because there is no longer compression across the failed circumferential welds, some portion of each vertical weld has to be intact to react the hoop load in the shroud due to pressure. Ten inches of weld ligament at the top of each vertical weld is assumed to be intact. (Note that the weld ligament is assumed to be at the top because this results in the highest bending stresses in the weld ligament from the seismic bumper load.)

Loads are applied to the model in the same way as described in the main body of this calculation. One case is run with the MSLB differential pressure of 19.0 psi. Another case is run with a seismic fuel load of 40 kips. The bumper load is also 40 kips because the horizontal welds are not capable of carrying shear and will not transfer an additional lateral load through the shroud sections to the bumper. Stresses are combined by direct summation.

The ANSYS runs have been performed with the following key stress results. The maximum membrane plus bending stress at the bumper contact is 52 ksi compared to an allowable stress of 60 ksi. The maximum membrane stress in the intact vertical weld is about 7 ksi compared to an allowable stress of 40 ksi, and the maximum membrane plus bending stress in the shroud section is 28 ksi compared to an allowable stress of 60 ksi. The stress contour for the evaluation is shown in Figure A-1. The ANSYS analyses are documented References A-1 and A-2.

A.4 REFERENCES

A-1. ANSYS Output File "press8.out", 7/10/98, 10:13.

A-2. ANSYS Output File "slbsse8a.out", 7/10/98, 08:57.

Figure A-1. Stress Contour for the MSLB + SSE Load Case with Only H5 and H6A Failed



APPENDIX B

Leakage Path Flow Area During Normal Operation

B.1 PURPOSE

The purpose of this appendix is to determine the maximum leakage path flow area through a flawed vertical weld during normal operation. This appendix accounts for the effect of the core plate wedges installed between the core plate and the shroud shell between H5 and H6A.

B.2 RESULTS

The leakage path flow area during normal operating conditions through a single, fully-cracked vertical weld in the H5/H6A shroud segment is 0.495 in^2.

B.3 CALCULATION

The maximum leakage path flow area through a flawed vertical weld in the H5/H6A shroud segment is determined using the finite element model developed in this calculation. As in Section 3, the vertical weld is assumed to be completely cracked. During normal operating conditions, the only load that opens the cracked weld is differential pressure. The normal operating differential pressure is 4.34 psi (Reference 5). With this load, using the same boundary conditions as in Section 3 (except that the core support ring is assumed to be intact), the finite element model is re-solved to examine the displaced shape at the crack.

The total leakage path flow area can be determined by examining the circumferential displacement of selected nodes relative to the corresponding nodes on the opposite side of the crack face. The trapezoidal flow area between the nodes is determined using the following formula:

$$A_i = (Z_{i+1} - Z_i)\,[\,(UY_{Right,\,i} - UY_{Left,\,i}) + (UY_{Right,\,i+1} - UY_{Left,\,i+1})\,]\,/\,2$$

Where:

A_i	=	Flow area between the i^{th} and $(i+1)^{th}$ nodes (in^2)
Z	=	Distance along vertical weld (in)
UY_{Left}	=	Circumferential displacement of node on left crack face (in)
UY_{Right}	=	Circumferential displacement of node on right crack face (in)

The total leakage path flow area is the sum of the individual flow areas between adjacent nodes. The calculation is summarized in Table B-1. The ANSYS analysis is documented in Reference B-1.

B.4 REFERENCES

B-1. ANSYS Output File "press1.out", 5/22/98, 12:53.

Table B-1. **Total Leakage Path Flow Area**

Z (in)	Left Crack Face		Right Crack Face		Flow Area (in²)
	Node Number	UY_{Left} (in)	Node Number	UY_{Right} (in)	
0.00	17732	-0.00004	17732	-0.00004	---
4.00	17734	-0.00008	17734	-0.00008	0.000
7.00	17751	-0.00125	17851	0.00120	0.004
10.87	17796	-0.00233	17939	0.00229	0.014
14.16	17795	-0.00318	17938	0.00316	0.018
16.95	17794	-0.00383	17937	0.00382	0.020
19.32	17793	-0.00432	17936	0.00433	0.019
21.34	17792	-0.00470	17935	0.00471	0.018
23.05	17791	-0.00498	17934	0.00500	0.017
24.50	17738	-0.00519	17833	0.00522	0.015
26.00	17742	-0.00539	17827	0.00542	0.016
27.44	17757	-0.00554	17865	0.00558	0.016
29.55	17758	-0.00573	17866	0.00577	0.024
32.64	17759	-0.00590	17867	0.00596	0.036
37.18	17760	-0.00592	17868	0.00599	0.054
43.85	17761	-0.00547	17869	0.00553	0.076
53.64	17762	-0.00391	17870	0.00402	0.093
68.00	17756	-0.00004	17756	-0.00004	0.057
				Total Leakage Path Flow Area:	0.495

Appendix C

ANSYS Input Files

Recirc Line Break Evaluation, H4-H6b model, 8 wedges


```
/batch,list
/filnam,recsse8a
/title,Oyster Creek Shroud Vertical Weld Evaluation
/psearch,/d0/blane/wedges/
/prep7

fapplied=40000          ! applied load from core (lbs)
flateral=34000+41300    ! lateral load from shroud (lbs)
thf=165                 ! angle of applied load
thb0=75                 ! 1/2 angle over which bumpers participate
thw0=180                ! 1/2 angle over which wedges participate
hintact1=-.1            ! intact length at bottom of weld
hintact2=-.1            ! intact length at top of weld

modl46b8
!resume,mesh,db,/d0/blane/wedges/

mp,ex,1,25.4e6
mp,nuxy,1,0.3
mp,dens,1,0.29
mp,ex,2,25.4e6
mp,nuxy,2,0.3
mp,dens,2,0.001

bnds46b

fini
/solu
antype,static
solve
save
```

Main Steam Line Break Evaluation, H4-H6b model, 8 wedges


```
/batch,list
/filnam,slbsse8a
/title,Oyster Creek Shroud Vertical Weld Evaluation
/psearch,/d0/blane/wedges/
/prep7

fapplied=40000          ! applied load from core (lbs)
flateral=34000          ! lateral load from shroud (lbs)
thf=165                 ! angle of applied load
thb0=75                 ! 1/2 angle over which bumpers participate
thw0=180                ! 1/2 angle over which wedges participate
hintact1=-.1            ! intact length at bottom of weld
hintact2=-.1            ! intact length at top of weld

modl46b8
!resume,mesh,db,/d0/blane/wedges/

mp,ex,1,25.4e6
mp,nuxy,1,0.3
mp,dens,1,0.29
mp,ex,2,25.4e6
mp,nuxy,2,0.3
mp,dens,2,0.001

bnds46b

fini
/solu
antype,static
solve
save
```

Main Steam Line Break Evaluation, H5-H6a model, 8 wedges


```
/batch,list
/filnam,slbsse8a
/title,Oyster Creek Shroud Vertical Weld Evaluation
/psearch,/d0/blane/wedges/
/prep7

fapplied=40000          ! applied load from core (lbs)
flateral=0              ! lateral load from shroud (lbs)
thf=165                 ! angle of applied load
thb0=75                 ! 1/2 angle over which bumpers participate
thw0=180                ! 1/2 angle over which wedges participate
hintact1=-.1            ! intact length at bottom of weld
hintact2=10             ! intact length at top of weld

modl56a8
!resume,mesh,db,/d0/blane/wedges/

mp,ex,1,25.4e6
mp,nuxy,1,0.3
mp,dens,1,0.29

bnds56a

fini
/solu
antype,static
solve
save
```

Internal pressure evaluation, H4-H6b model, 8 wedges


```
/batch,list
/filnam,press8
/title,Oyster Creek Shroud Vertical Weld Evaluation
/psearch,/d0/blane/wedges/
/prep7

thf=165
thb0=75
thw0=180
hintact1=-.1             ! intact length at bottom of weld
hintact2=-.1             ! intact length at top of weld

modl46b8
!resume,mesh,db,/d0/blane/wedges/

mp,ex,1,25.4e6
mp,nuxy,1,0.3
mp,dens,1,0.29
mp,ex,2,25.4e6
mp,nuxy,2,0.3
mp,dens,2,0.001

prss46

fini
/solu
antype,static

csys,1          ! dp=4.34
cmsel,s,shroud
aslv
lsla
lsel,r,loc,x,r1-.01,r1+.01
asll,s,1
sfa,all,,pres,4.34
allsel
csys,0
solve

csys,1          ! dp=19.0
cmsel,s,shroud
aslv
lsla
lsel,r,loc,x,r1-.01,r1+.01
asll,s,1
sfa,all,,pres,19.0
allsel
csys,0
solve

save
```

Internal pressure evaluation, H5-H6a model, 8 wedges


```
/batch,list
/filnam,press8
/title,Oyster Creek Shroud Vertical Weld Evaluation
/psearch,/d0/blane/wedges/
/prep7

thf=165
thb0=75
thw0=180
hintact1=-.1            ! intact length at bottom of weld
hintact2=10             ! intact length at top of weld

mod156a8
!resume,mesh,db,/d0/blane/wedges/

mp,ex,1,25.4e6
mp,nuxy,1,0.3
mp,dens,1,0.29

prss56a

fini
/solu
antype,static

!csys,1          ! dp=4.34
!cmsel,s,shroud
!aslv
!lsla
!lsel,r,loc,x,r1-.01,r1+.01
!asll,s,1
!sfa,all,,pres,4.34
!allsel
!solve

csys,1           ! dp=19.0
cmsel,s,shroud
aslv
lsla
lsel,r,loc,x,r1-.01,r1+.01
asll,s,1
sfa,all,,pres,19.0
allsel
csys,0
solve

save
```

Geometry for H4-H6b model, 8 wedges


```
!       oyster creek shroud (h4 to h6b)

/prep7
*afun,deg

! shroud dimensions

r1=176/2                ! ID
r2=r1+1.5               ! OD
r3=r2-8.75              ! ID of core support ring

z6b=0                   ! h6B weld elevation
z6a=z6b+4               ! h6A weld elevation
z5=z6a+64               ! h5 weld elevation
z4=z5+89.875            ! h4 weld elevation

th1=164                 ! vertical weld 1
th2=345                 ! vertical weld 2

! bumper dimensions

zb1=z6b                 ! bottom of lower bumper
zb2=z6a+3               ! top of lower bumper
dbw=2                   ! width of bumper
drb=4                   ! radial thickness
thbw=2*asin(dbw/(2*r2)) ! angular width of bumper

nbumpers=7              ! number of bumpers
*dim,thb,array,nbumpers ! bumper angular locations
thb(1)=10,70,100,160
thb(5)=220,280,310

! wedge dimensions

zw1=z6a+20.5            ! bottom of wedge
zw2=z6a+22              ! top of wedge
dww=4                   ! width of wedge
drw=2.4                 ! radial thickness
thww=2*asin(dww/(2*r1)) ! angular width of wedge

nwedges=8               ! number of wedges
*dim,thw,array,nwedges  ! wedge angular locations
thw(1)=24-thww/2,60+thww/2,96-thww/2,132+thww/2
thw(5)=204+thww/2,240-thww/2,276+thww/2,312-thww/2

!       Geometry

wprota,0,90
rectng,r1,r2,z6a,z5
rectng,r1,r2,z6a,zb2
rectng,r1,r2,zw1,zw2
*if,hintact1,gt,0,then
  rectng,r1,r2,z6a,z6a+hintact1
*endif
*if,hintact2,gt,0,then
  rectng,r1,r2,z5-hintact2,z5
*endif
aovlap,all
rectng,r1,r2,z5,z4
rectng,r3,r2,z6b,z6a
wpstyl,defa
numcmp,all

thcurr=0
windx=1
bindx=1
*dim,thbl,array,nbumpers*2+1
*dim,thwl,array,nwedges*2+1
*do,ii,1,nbumpers,1
  thbl(2*ii-1)=thb(ii)-thbw/2
  thbl(2*ii)=thb(ii)+thbw/2
*enddo
*do,ii,1,nwedges,1
  thwl(2*ii-1)=thw(ii)-thww/2
  thwl(2*ii)=thw(ii)+thww/2
*enddo
thbl(2*nbumpers+1)=1000
thwl(2*nwedges+1)=1000
k,1000,0,0,z6a
k,1001,0,0,z5
```

```
local,11,1,0,0,0,0,180,0
csys,11
*do,ii,1,2*(nbumpers+nwedges),1
  *if,thbl(bindx),lt,thwl(windx),then
    asel,s,loc,y,thcurr-.01,thcurr+.01
    *if,thbl(bindx)-thcurr,gt,5,then
      vrotat,all,,,,,,1001,1000,thbl(bindx)-thcurr,2
    *else
      vrotat,all,,,,,,1001,1000,thbl(bindx)-thcurr,1
    *endif
    thcurr=thbl(bindx)
    bindx=bindx+1
  *else
    asel,s,loc,y,thcurr-.01,thcurr+.01
    *if,thwl(windx)-thcurr,gt,5,then
      vrotat,all,,,,,,1001,1000,thwl(windx)-thcurr,2
    *else
      vrotat,all,,,,,,1001,1000,thwl(windx)-thcurr,1
    *endif
    thcurr=thwl(windx)
    windx=windx+1
  *endif
*enddo
asel,s,loc,y,thcurr-.01,thcurr+.01
vrotat,all,,,,,,1001,1000,360-thcurr,2

csys,0
vsel,s,loc,z,z6b,z6a
aslv
lsla
ksll
nummrg,kp
vsel,s,loc,z,z6a,z5
aslv
lsla
ksll
nummrg,kp
vsel,s,loc,z,z5,z4
aslv
lsla
ksll
nummrg,kp
numcmp,all

!       cut the shroud at the vertical welds

wpstyl,defa
wprota,-th1,90
csys,11
vsel,s,loc,y,th1-10,th1+10
vsbw,all,sepo
wpstyl,defa
wprota,-th2,90
vsel,s,loc,y,th2-30,th2+10
vsbw,all,sepo
wpstyl,defa

! join shroud at the vertical welds as appropriate

csys,11
vsel,s,loc,z,-z5,-z6a
asel,s,loc,y,th1
asel,a,loc,y,th2
aslv,r
lsla
lsel,r,loc,z,-z6a+.01,-(z6a+hintact1)
asll,s,1
ksll
nummrg,kp
asel,s,loc,y,th1
asel,a,loc,y,th2
aslv,r
lsla
lsel,r,loc,z,-(z5-hintact2),-z5-.01
asll,s,1
ksll
nummrg,kp
*if,hintact1,ge,0,then
  vsel,s,loc,z,-z6b,-z6a
  asel,s,loc,y,th1
  asel,a,loc,y,th2
```

```
  aslv,r
  lsla
  ksll
  nummrg,kp
*endif
vsel,s,loc,z,-z4,-z5
asel,s,loc,y,th1
asel,a,loc,y,th2
aslv,r
lsla
ksll
nummrg,kp

! join shroud at horizontal welds as appropriate

csys,1
vsel,none
asel,none
*if,hintact1,lt,0,then
  lsel,s,loc,x,r2-.01,r2+.01
  lsel,r,loc,y,-th1,-th2
  lsel,r,loc,z,z6a
  ksll
  nummrg,kp
  lsel,inve
  lsel,r,loc,x,r2-.01,r2+.01
  lsel,r,loc,z,z6a
  ksll
  nummrg,kp
*else
  lsel,s,loc,x,r2-.01,r2+.01
  lsel,r,loc,z,z6a
  ksll
  nummrg,kp
*endif
lsel,s,loc,x,r2-.01,r2+.01
lsel,r,loc,z,z5
ksll
nummrg,kp

csys,0
allsel

cm,shroud,volu

!        Mesh

et,1,solid45
type,1
eshape,2

csys,1
vsel,s,loc,z,z6a,z5
aslv
lsla
lsel,r,loc,x,r1+.01,r2-.01
lesize,all,,,4
vsel,s,loc,z,z5,z4
aslv
lsla
lsel,r,loc,x,r1+.01,r2-.01
lesize,all,,,1
allsel

csys,1
ksel,all
kesize,all,15
ksel,s,loc,z,zw1-.1,zw2+.1
kesize,all,1.5
ksel,s,loc,z,zb1-.1,zb2+.1
kesize,all,4
lsel,s,loc,z,zb2+.1,zw1-.1
lsel,a,loc,z,zw2+.1,z5-.1
lesize,all

csys,11
allsel
kesize,all,8
ksel,s,loc,y,thf-thw0,thf+thw0
kesize,all,4
*do,ii,1,nwedges,1
```

```
  vsel,s,loc,y,thw(ii)-thww/2,thw(ii)+thww/2
  vsel,r,loc,z,-zw1,-zw2
  vsel,r,loc,y,thf-thw0,thf+thw0
  aslv
  lsla
  ksll
  kesize,all,1.5
*enddo
csys,1
ksel,s,loc,z,z5-1,z4+.1
kesize,all,15
allsel

csys,1
allsel
vsel,s,loc,z,z6b,z5
mat,1
vmesh,all
vsel,s,loc,z,z5,z4
mat,2
vmesh,all

!       add the wedges

vsel,none
wpstyl,defa
wprota,0,180,0
*do,ii,1,nwedges,1
  cylind,r1-drw,r1,-zw1,-zw2,thw(ii)-thww/2,thw(ii)+thww/2
*enddo
wpstyl,defa
cm,wedges,volu
csys,0
allsel

!       add the bumpers

vsel,none
wpstyl,defa
wprota,0,180,0
*do,ii,1,nbumpers,1
  cylind,r2,r2+drb,-zb1,-zb2,thb(ii)-thbw/2,thb(ii)+thbw/2
*enddo
wpstyl,defa
cm,bumpers,volu
csys,0
allsel

esize,2
vmesh,all

allsel
csys,0
```

Geometry for H5 - H6a model with 8 wedges


```
!       oyster creek shroud (h5 to h6a)

/prep7
*afun,deg

! shroud dimensions

r1=176/2                   ! ID
r2=r1+1.5                  ! OD

z6a=0                      ! h6A weld elevation
z5=z6a+64                  ! h5 weld elevation

th1=164                    ! vertical weld 1
th2=345                    ! vertical weld 2

! bumper dimensions

zb2=3                      ! top of bumper elevation
dbw=2                      ! width of bumper
drb=4                      ! radial thickness
thbw=2*asin(dbw/(2*r2)) ! angular width of bumper

nbumpers=7                 ! number of bumpers
*dim,thb,array,nbumpers ! bumper angular locations
thb(1)=10,70,100,160
thb(5)=220,280,310

! wedge dimensions

zw1=20.5                   ! bottom of wedge elevation
zw2=22                     ! top of wedge elevation
dww=4                      ! width of wedge
drw=2.4                    ! radial thickness
thww=2*asin(dww/(2*r1)) ! angular width of wedge

nwedges=8                  ! number of wedges
*dim,thw,array,nwedges  ! wedge angular locations
thw(1)=24-thww/2,60+thww/2,96-thww/2,132+thww/2
thw(5)=204+thww/2,240-thww/2,276+thww/2,312-thww/2

!       Geometry

wprota,0,90
rectng,r1,r2,z6a,z5
rectng,r1,r2,z6a,zb2
rectng,r1,r2,zw1,zw2
*if,hintact1,gt,0,then
  rectng,r1,r2,z6a,z6a+hintact1
*endif
*if,hintact2,gt,0,then
  rectng,r1,r2,z5-hintact2,z5
*endif
wpstyl,defa
aovlap,all
numcmp,all

thcurr=0
windx=1
bindx=1
*dim,thbl,array,nbumpers*2+1
*dim,thwl,array,nwedges*2+1
*do,ii,1,nbumpers,1
  thbl(2*ii-1)=thb(ii)-thbw/2
  thbl(2*ii)=thb(ii)+thbw/2
*enddo
*do,ii,1,nwedges,1
  thwl(2*ii-1)=thw(ii)-thww/2
  thwl(2*ii)=thw(ii)+thww/2
*enddo
thbl(2*nbumpers+1)=1000
thwl(2*nwedges+1)=1000
k,1000,0,0,z6a
k,1001,0,0,z5

local,11,1,0,0,0,0,180,0
csys,11
*do,ii,1,2*(nbumpers+nwedges),1
  *if,thbl(bindx),lt,thwl(windx),then
    asel,s,loc,y,thcurr-.01,thcurr+.01
    *if,thbl(bindx)-thcurr,gt,5,then
```

```
      vrotat,all,,,,,,1001,1000,thbl(bindx)-thcurr,2
    *else
      vrotat,all,,,,,,1001,1000,thbl(bindx)-thcurr,1
    *endif
    thcurr=thbl(bindx)
    bindx=bindx+1
  *else
    asel,s,loc,y,thcurr-.01,thcurr+.01
    *if,thwl(windx)-thcurr,gt,5,then
      vrotat,all,,,,,,1001,1000,thwl(windx)-thcurr,2
    *else
      vrotat,all,,,,,,1001,1000,thwl(windx)-thcurr,1
    *endif
    thcurr=thwl(windx)
    windx=windx+1
  *endif
*enddo
asel,s,loc,y,thcurr-.01,thcurr+.01
vrotat,all,,,,,,1001,1000,360-thcurr,2

csys,0
allsel
nummrg,all
numcmp,all

!       cut the shroud at the vertical welds

wpstyl,defa
wprota,-th1,90
csys,11
vsel,s,loc,y,th1-10,th1+10
vsbw,all,sepo
wpstyl,defa
wprota,-th2,90
vsel,s,loc,y,th2-30,th2+10
vsbw,all,sepo
wpstyl,defa

! join shroud at the vertical welds as appropriate

csys,11
asel,s,loc,y,th1
asel,a,loc,y,th2
lsla
lsel,r,loc,z,-z6z+.01,-(z6a+hintact1)
asll,s,1
ksll
nummrg,kp
asel,s,loc,y,th1
asel,a,loc,y,th2
lsla
lsel,r,loc,z,-(z5-hintact2),-z5-.01
asll,s,1
ksll
nummrg,kp

csys,0
allsel
wpstyl,defa

cm,shroud,volu

!       Mesh

et,1,solid45
type,1
eshape,2

ksel,all
csys,11
vsel,s,loc,y,thf-thw0,thf+thw0
kesize,all,4
ksel,inve
kesize,all,8
*do,ii,1,nwedges,1
  vsel,s,loc,y,thw(ii)-thww/2,thw(ii)+thww/2
  vsel,r,loc,z,-zw1,-zw2
  vsel,r,loc,y,thf-thw0,thf+thw0
  aslv
  lsla
  ksll
```

```
  kesize,all,1.5
*enddo
csys,1
lsel,s,loc,x,r1+.01,r2-.01
lesize,all,,,4

allsel
mat,1
vmesh,all

!       add the wedges

vsel,none
wpstyl,defa
wprota,0,180,0
*do,ii,1,nwedges,1
  cylind,r1-drw,r1,-zw1,-zw2,thw(ii)-thww/2,thw(ii)+thww/2
*enddo
wpstyl,defa
cm,wedges,volu
csys,0
allsel

!       add the bumpers

vsel,none
wpstyl,defa
wprota,0,180,0
*do,ii,1,nbumpers,1
  cylind,r2,r2+drb,-z6a,-zb2,thb(ii)-thbw/2,thb(ii)+thbw/2
*enddo
wpstyl,defa
cm,bumpers,volu
csys,0
allsel

esize,2
vmesh,all
```

Boundary conditions for lateral load cases, H4-H6b models


```
!       boundary condition macro

! couple wedges and shroud

csys,11
*do,ii,1,nwedges,1
  asel,s,loc,x,r1-.01,r1+.01
  asel,r,loc,y,thw(ii)-thww/2,thw(ii)+thww/2
  asel,r,loc,z,-zw1,-zw2
  nsla,s,1
  cpintf,all,.1
*enddo
allsel
csys,0

! couple bumpers and shroud

csys,11
*do,ii,1,nbumpers,1
  asel,s,loc,x,r2-.01,r2+.01
  asel,r,loc,y,thb(ii)-thbw/2,thb(ii)+thbw/2
  asel,r,loc,z,-zb1,-zb2
  nsla,s,1
  cpintf,all,.1
*enddo
allsel
csys,0

! create rigid core plate and apply load

*get,nmax,node,,num,max
n,nmax+1,0,0,(zw1+zw2)/2
et,2,mass21
type,2
r,1,.01
real,1
e,nmax+1

csys,11
nsel,none
cm,ntmp,node
*do,ii,1,nwedges,1
  nsel,all
  nsel,s,node,,node(r1-drw,thw(ii),-(zw1+zw2)/2)
  cmsel,a,ntmp
  cm,ntmp,node
*enddo
nsel,a,node,,nmax+1
cerigid,nmax+1,all,uxyz
csys,0
allsel

d,nmax+1,uz,0
d,nmax+1,rotx,0
d,nmax+1,roty,0
f,nmax+1,fx,fapplied*cos(-thf)
f,nmax+1,fy,fapplied*sin(-thf)

! apply constraint at lower bumpers

csys,11
nsel,none
cm,ntmp,node
*do,ii,1,nbumpers,1
  nsel,all
  nsel,s,node,,node(r2+drb,thb(ii),-zb1/2)
  cmsel,a,ntmp
  cm,ntmp,node
*enddo
nsel,r,loc,y,thf-thb0,thf+thb0
cm,ntmp,node

n1=node(r2+drb,thf,-zb1/2)       ! leading bumper
nrotat,n1
d,n1,ux,0
deltlead=.375/cos(thf-ny(n1))

nsel,u,node,,n1          ! trailing bumpers
cm,ntmp,node
*get,nnum,node,,count
n1=0
```

```
*do,i1,1,nnum,1
  n1=ndnext(n1)
  nrotat,n1
!  d,n1,ux,.375-deltlead*cos(thf-ny(n1))
  d,n1,ux,0
*enddo

csys,0
allsel

! apply constraint at intermediate bumpers
*if,skip,eq,1,then

*dim,thbmid,array,3
thbmid(1)=100,220,350
csys,11
*do,ii,1,3,1
  nsel,s,loc,x,r2-.01,r2+.01
  nsel,r,loc,y,thbmid(ii)-thbw/2,thbmid(ii)+thbw/2
  nsel,r,loc,z,-z5
  nsel,r,loc,y,thf-thb0,thf+thb0
  nrotat,all
  d,all,ux,0
  nsel,s,loc,x,r2-.01,r2+.01
  nsel,r,loc,y,thbmid(ii)-thbw/2,thbmid(ii)+thbw/2
  nsel,r,loc,z,-z4
  nsel,r,loc,y,thf-thb0,thf+thb0
  nrotat,all
  d,all,ux,0
*enddo
allsel
csys,0
*endif

! apply additional lateral load as an acceleration

cmsel,s,shroud
vsel,r,loc,z,z6b,z5
vsum
*get,voltot,volu,,volu
*get,voldens,dens,1
mass1=voltot*voldens
vsel,inve
vsum
*get,voltot,volu,,volu
*get,voldens,dens,2
mass2=voltot*voldens
accel=flateral/(mass1+mass2)
acel,-accel*cos(-thf),-accel*sin(-thf),0

csys,0
allsel
```

Boundary conditions for lateral load cases, H5-H6a model


```
!       boundary condition macro

! couple wedges and shroud

csys,11
*do,ii,1,nwedges,1
  asel,s,loc,x,r1-.01,r1+.01
  asel,r,loc,y,thw(ii)-thww/2,thw(ii)+thww/2
  asel,r,loc,z,-zw1,-zw2
  nsla,s,1
  cpintf,all,.1
*enddo
allsel
csys,0

! couple bumpers and shroud

csys,11
*do,ii,1,nbumpers,1
  asel,s,loc,x,r2-.01,r2+.01
  asel,r,loc,y,thb(ii)-thbw/2,thb(ii)+thbw/2
  asel,r,loc,z,-zb1,-zb2
  nsla,s,1
  cpintf,all,.1
*enddo
allsel
csys,0

! create rigid core plate and apply load

*get,nmax,node,,num,max
n,nmax+1,0,0,(zw1+zw2)/2
et,2,mass21
type,2
r,1,.01
real,1
e,nmax+1

csys,11
nsel,none
cm,ntmp,node
*do,ii,1,nwedges,1
  nsel,all
  nsel,s,node,,node(r1-drw,thw(ii),-(zw1+zw2)/2)
  cmsel,a,ntmp
  cm,ntmp,node
*enddo
nsel,a,node,,nmax+1
cerigid,nmax+1,all,uxyz
csys,0
allsel

d,nmax+1,uz,0
d,nmax+1,rotx,0
d,nmax+1,roty,0
f,nmax+1,fx,fapplied*cos(-thf)
f,nmax+1,fy,fapplied*sin(-thf)

! apply constraint at lower bumpers

csys,11
nsel,none
cm,ntmp,node
*do,ii,1,nbumpers,1
  nsel,all
  nsel,s,node,,node(r2+drb,thb(ii),-zb1/2)
  cmsel,a,ntmp
  cm,ntmp,node
*enddo
nsel,r,loc,y,thf-thb0,thf+thb0
cm,ntmp,node

n1=node(r2+drb,thf,-zb1/2)        ! leading bumper
nrotat,n1
d,n1,ux,0
deltlead=.375/cos(thf-ny(n1))

nsel,u,node,,n1          ! trailing bumpers
cm,ntmp,node
*get,nnum,node,,count
n1=0
```

```
*do,ii,1,nnum,1
  nl=ndnext(nl)
  nrotat,nl
!  d,nl,ux,.375-deltlead*cos(thf-ny(nl))
  d,nl,ux,0
*enddo

csys,0
allsel

! apply constraint at intermediate bumpers
*if,skip,eq,1,then

*dim,thbmid,array,3
thbmid(1)=100,220,350
csys,11
*do,ii,1,3,1
  nsel,s,loc,x,r2-.01,r2+.01
  nsel,r,loc,y,thbmid(ii)-thbw/2,thbmid(ii)+thbw/2
  nsel,r,loc,z,-z5
  nsel,r,loc,y,thf-thb0,thf+thb0
  nrotat,all
  d,all,ux,0
  nsel,s,loc,x,r2-.01,r2+.01
  nsel,r,loc,y,thbmid(ii)-thbw/2,thbmid(ii)+thbw/2
  nsel,r,loc,z,-z4
  nsel,r,loc,y,thf-thb0,thf+thb0
  nrotat,all
  d,all,ux,0
*enddo
allsel
csys,0
*endif

! apply additional lateral load as an acceleration

cmsel,s,shroud
vsel,r,loc,z,z6a,z5
vsum
*get,voltot,volu,,volu
*get,voldens,dens,1
mass1=voltot*voldens
accel=flateral/mass1
acel,-accel*cos(-thf),-accel*sin(-thf),0

csys,0
allsel
```

Boundary conditions for pressure evaluation, H4-H6b model


```
!    boundary condition macro for applied pressure case

! couple wedges and shroud

csys,11
*do,ii,1,nwedges,1
  asel,s,loc,x,r1-.01,r1+.01
  asel,r,loc,y,thw(ii)-thww/2,thw(ii)+thww/2
  asel,r,loc,z,-zw1,-zw2
  nsla,s,1
  cpintf,all,.1
*enddo
allsel
csys,0

! couple bumpers and shroud

csys,11
*do,ii,1,nbumpers,1
  asel,s,loc,x,r2-.01,r2+.01
  asel,r,loc,y,thb(ii)-thbw/2,thb(ii)+thbw/2
  asel,r,loc,z,-zb1,-zb2
  nsla,s,1
  cpintf,all,.1
*enddo
allsel
csys,0

! restrain vertically at h4

csys,0
asel,s,loc,z,z4
nsla,s,1
d,all,uz,0
allsel

! restrain circumferentially and radially at h4

csys,11
nn1=node(r1,th1,-z4),
nn2=node(r1,th2,-z4),
nsel,s,node,,nn1
nsel,a,node,,nn2
nrotat,all
d,all,uy,all
allsel
d,nn2,ux,all

! add dummy node for adding loadcases

allsel
*get,nmax,node,,num,max
n,nmax+1,0,0,0
et,2,mass21
type,2
r,1,.001
real,1
e,nmax+1
cp,next,all,nmax+1,node(r1,0,0)

csys,0
allsel
```

Boundary conditions for pressure case, H5-H6a model


```
!     pressure boundary condition macro

! couple wedges and shroud

csys,11
*do,ii,1,nwedges,1
  asel,s,loc,x,r1-.01,r1+.01
  asel,r,loc,y,thw(ii)-thww/2,thw(ii)+thww/2
  asel,r,loc,z,-zw1,-zw2
  nsla,s,1
  cpintf,all,.1
*enddo
allsel
csys,0

! couple bumpers and shroud

csys,11
*do,ii,1,nbumpers,1
  asel,s,loc,x,r2-.01,r2+.01
  asel,r,loc,y,thb(ii)-thbw/2,thb(ii)+thbw/2
  asel,r,loc,z,-z6a,-zb2
  nsla,s,1
  cpintf,all,.1
*enddo
allsel
csys,0

*if,skip,eq,1,then
! apply constraint at lower bumpers

csys,11
nsel,none
cm,ntmp,node
*do,ii,1,nbumpers,1
  nsel,all
  nsel,s,node,,node(r2+drb,thb(ii),-z6a/2)
  cmsel,a,ntmp
  cm,ntmp,node
*enddo
nsel,r,loc,y,thf-thb0,thf+thb0
cm,ntmp,node

n1=node(r2+drb,thf,-z6a/2)       ! leading bumper
nrotat,n1
d,n1,ux,0
deltlead=.375/cos(thf-ny(n1))

nsel,u,node,,n1             ! trailing bumpers
cm,ntmp,node
*get,nnum,node,,count
n1=0
*do,ii,1,nnum,1
  n1=ndnext(n1)
  nrotat,n1
!  d,n1,ux,.375-deltlead*cos(thf-ny(n1))
  d,n1,ux,0
*enddo

csys,0
allsel

! apply constraint at intermediate bumpers
*dim,thbmid,array,3
thbmid(1)=100,220,350
csys,11
*do,ii,1,3,1
  nsel,s,loc,x,r2-.01,r2+.01
  nsel,r,loc,y,thbmid(ii)-thbw/2,thbmid(ii)+thbw/2
  nsel,r,loc,z,-z5
  nsel,r,loc,y,thf-thb0,thf+thb0
  nrotat,all
  d,all,all,0
*enddo
*endif

! displacement constraints
csys,1
nsel,s,loc,z,z5-.1,z5+.1
nrotat,all
d,all,uy,0
```

```
d,all,uz,0

! add dummy node and mass element for adding load cases
allsel
*get,nmax,node,,num,max
n,nmax+1,0,0,(zw1+zw2)/2
et,2,mass21
type,2
r,1,.01
real,1
e,nmax+1
cp,next,all,nmax+1,node(r1,0,0)

allsel
csys,0
```